
Abstract. The Ontology Web Language (OWL) is a language used for
the Semantic Web. OWL is based on Description Logics (DLs), a family
of logical formalisms for representing and reasoning about conceptual
and terminological knowledge. Among these, the logic ALC is a ground
DL used in many practical cases. Moreover, the Semantic Web appears
as a new field for the application of formal methods, that could be used
to increase its reliability. A starting point could be the formal verifica-
tion of satisfiability provers for DLs. In this paper, we present the PVS
specification of a prover for ALC , as well as the proofs of its termina-
tion, soundness and completeness. We also present the formalization of
the well–foundedness of the multiset relation induced by a well–founded
relation. This result has been used to prove the termination and the
completeness of the ALC prover.

1 Introduction

The goal of the presented work is the formal verification of satisfiability algo-
rithms for description logics (DLs), as a previous stage to the formal verification
of DLs reasoners. In particular, we describe in this paper a formal proof of the
well–known tableau algorithm for the ALC description logic in the PVS verifica-
tion system [19].

Description Logics [5] are a family of logics which can be used to represent
terminological and conceptual knowledge. Among these, the ground logic is the
ALC logic, introduced by Schmidt–Schauß and Smolka [24], who also developed
a tableau–like algorithm for testing satisfiability in it. ALC is the base for more
expressive logics as SHOIQ, obtained extending ALC in several expressive ways.
The importance of SHOIQ stems from the fact that it and its fragments are
used for reasoning in the semantic web [4]. Specifically, the fragment SHOIN
corresponds to the ontology language OWL–DL [13, 12], which was recommended
by the W3C as the standard web ontology language [6]. The fragment SHIQ is
the concept language supported by systems as FaCT++ and RACER [25, 10].

⋆ This research was partially funded by Spanish Ministry of Education and Science
under grant TIN2004–03884 and Feder funds

Many techniques have been proposed and investigated to obtain decision
procedures for DLs reasoning. Among these, tableau reasoning are the most
successful approach so far. In fact, DLs reasoners such as FaCT++ and RACER
are based on tableau algorithms.

We believe that the verification of reasoning systems for the SW poses a new
challenge for the application of formal methods. In this field, our research group
have carried out several works related to the formal verification of reasoning sys-
tems. Some examples are the verification of a generic framework for propositional
logic reasoning algorithms [14], Knuth–Bendix based reasoning [22], Buchberger
algorithm for polynomial–based logical reasoning [15, 16] and conceptual pro-
cessing in Formal Concept Analysis [3].

In this paper, we show a proof in PVS of correctness of the tableau algorithm
for ALC described in theoretical papers. The hardest part of this task is the
termination proof, for which we have extended the multiset library of PVS in
order to include well–foundedness of the multiset order relation. We use this
property and a measure function as it is described in [9] to prove the termination
of the algorithm.

The paper is structured as follows. Section 2 is devoted to describe the proof
of the well–foundedness of the multiset induced relation of a well–founded rela-
tion. Section 3 shows the syntax and semantics of the ALC logic and how we have
formalized it in the PVS language specification. Also, it contains the specification
in PVS of a prover for the ALC logic, with some details about its termination,
soundness and completeness proofs. Section 4 is devoted to explain how we have
constructed in PVS a concrete measure function. Finally, in the last section we
draw some conclusions and suggest some lines of future work.

Since the whole formalization consists of a large collection of PVS theorems
and definitions, we give an overview presenting the main results and a sketch of
how the pieces fit together. We necessarily omit many details that, we expect,
can be inferred from the context. We urge the interested reader to consult the
whole PVS theory developed, which is widely documented and avaliable at http:
//www.cs.us.es/~mjoseh/alc/.

2 Well–founded ordering of multisets

Multisets (bags) over a given set T are sets that admit multiple occurrences of
elements taken from T . Dershowitz and Manna [9] prove that every well–founded
relation on T induces a well–founded relation on the set of finite multisets over
T . They also show how the multiset ordering is used to prove the termination of
production systems, programs defined in term of rewriting rules. In our work, we
use this method to prove the termination and completeness of the ALC reasoner.

The goal of this section is to show our extension of the PVS multiset library,
to include the formalization of the Dershowitz and Manna theorem. The proof
formalized in the PVS system is based on the proof described by T. Nipkow
[17]. It uses a characterization of well–foundedness based on the notion of well–
founded part of a set, and the transitive closure of a well–founded binary relation.

2.1 An alternative characterization of well–foundedness

The definition of well–foundedness included in the PVS prelude is the usual one:
a relation < on T is well–founded if every non-empty subset of T has a minimal
element (w.r.t. <). Nevertheless, the proof of well-foundedness of the multiset
orderings that we have formalized is based on an alternative definition given by
P. Aczel [1]. Let us start describing the PVS proof of the equivalence between
Aczel’s definition and the usual one.

Given a binary well–founded relation < defined on a set T , the well–founded
part of T with respect to <, denoted as W (T, <), is the smallest subset of T
closed under the set of rules (∀a ∈ T )[(∀y < a)[y ∈ W (T, <)] → a ∈ W (T, <)].
In our PVS definition of W (T, <), both the type T and the relation < are
introduced as theory parameters. Also, we use the support that provides PVS
for constructing inductive definitions of sets or predicates.

[T: TYPE+, <: pred[[T,T]]]: THEORY

well_founded_part(x): INDUCTIVE bool =

FORALL y: y < x IMPLIES well_founded_part(y)

This way, the above inductive definition of the well–founded part generates,
automatically, the following induction axioms, which allow to prove properties
by induction on the defined set:

– Weak induction axiom for the well–founded part:

(∀x)[(∀y)[y < x → P (y)] → P (x)]

(∀x)[x ∈ W (T, <) → P (x)]

– Induction axiom for the well–founded part

(∀x)[(∀y)[y < x → y ∈ W (T, <) ∧ P (y)] → P (x)]

(∀x)[x ∈ W (T, <) → P (x)]

The following theorem characterizes the well–foundedness of a relation by
means of its well–founded part.

Theorem 1. (T, <) is well–founded if and only if W (T, <) = T .

well_founded_part_nsc: THEOREM

well_founded?[T](<) IFF (FORALL x: well_founded_part(x))

The PVS proof of the necessary condition is carried out by induction on (T, <)
with respect to the predicate x ∈ W (T, <). On the other hand, the sufficient
condition is proved using the weak induction axiom for the well–founded part of
T with respect to <.

Additionally, the PVS theory about well–foundedness that we have devel-
oped also includes alternative sufficient conditions of well–foundedness, like the
following embedding lemma.

Lemma 1. If f : (S, <′) → (T, <) is monotone and (T, <) is well–founded, then

(S, <′) is well–founded.

We prove it using the definition of well–foundedness based in the notion of
minimal element.

2.2 Well–foundedness of the transitive closure

The transitive closure of a binary relation on T , <, is the smallest relation <+

such that

(∀x, y ∈ T )[(x < y ∨ (∃z)[x <+ z ∧ z < y]) → x <+ y]

tr_cl(<)(x,y): INDUCTIVE bool =

x < y OR EXISTS z: tr_cl(<)(x,z) AND z < y

The main result about well–foundedness and transitive closure is the follo-
wing:

Theorem 2. If (T, <) is well–founded, then (T, <+) is well–founded.

well_founded_cl_tr: THEOREM

well_founded?[T](<) IMPLIES well_founded?[T](tr_cl(<))

To prove it, by theorem 1, it is sufficient to prove that W (T, <) ⊆ W (T, <+). We
prove this in PVS using the weak induction axiom generated by the definition
of W (T, <).

In a similar way, the reflexive transitive closure of a relation, rtr_cl(<), has
been defined and their main properties have been proved in PVS.

2.3 Well–founded multiset relations (in PVS)

In order to specify in PVS the multiset relations, we have used the PVS library
about bags 1. In this library, a multiset (bag) of elements in T is represented
by means of a function with domain T and range N. Let us start showing the
specification of the bag and finite_bag types, and also the specifications of
the basic operations insert and plus, included in it (in the following, we will
denote both as ⊎)

bag: TYPE = [T -> nat]

insert(x,b): bag = (LAMBDA t: IF x = t THEN b(t) + 1 ELSE b(t) ENDIF)

plus(a,b) : bag = (LAMBDA t: a(t) + b(t))

bag_to_set(b): set[T] = {t: T | b(t) > 0}

is_finite(b): bool = is_finite(bag_to_set(b))

finite_bag: TYPE = {b: bag | is_finite(b)}

1 Available at http://shemesh.larc.nasa.gov/fm/ftp/larc/PVS-library/pvslib.

html

Let < be a relation in T and M(T ) the set of finite multisets over T . The mul-

tiset relation induced by < on M(T ) is the relation <mult defined as: N <mult M
if there exist multisets M0, K1, K2 ∈ M(T ) such that K1 6= ∅, M = M0 ⊎ K1,
N = M0 ⊎ K2 and (∀a)[a ∈ K2 → (∃b)[b ∈ K1 ∧ a < b]]. An alternative defini-
tion could be obtained if instead of replacing a multiset K1 of elements in M , a
single element b ∈ M were replaced by a multiset of smaller elements (w.r.t. <).
That is, we could define the following multiset reduction relation denoted as <1:
N <1 M if there exists multisets M0, K2 and b ∈ M such that M = M0 ⊎ {b},
N = M0⊎K2 and (∀a)[a ∈ K2 → a < b]. It can be proved that if < is transitive,
then <mult is the transitive closure of <1. Therefore, by theorem 2, proving the
well–foundedness of <mult amounts to proving that <1 is well–founded, provided
that < is transitive.

We specify the relations <1 and <mult in PVS by less_1 and less_mult,
respectively, as follows

less(K,a): bool = FORALL b: member(b,K) IMPLIES b < a

less_1(N,M): bool =

EXISTS M_0,a,K: M = insert(a,M_0) AND N = plus(M_0,K) AND less(K,a)

less_mult(N,M): bool =

EXISTS M_0,K1,K2: nonempty_bag?(K1) AND M = plus(M_0, K1) AND

N = plus(M_0, K2) AND

FORALL a: member(a,K2) IMPLIES

EXISTS b: member(b,K1) AND a < b

Then, we stablish in PVS the following result, ensuring that the relation <1

is well–founded on M(T ).

Lemma 2. Let < be a well–founded relation on T . Then <1 is a well–founded

relation on M(T ).

wf_less_1: THEOREM well_founded?[finite_bag[T]](less_1)

To prove this lemma, using theorem 1, it is sufficient to prove that M(T ) ⊆
W (M(T ), <1). The PVS proof is carried out by induction on finite multisets,
according to the following scheme:

P (∅) ∧ (∀a)(∀M)[P (M) → P (M ⊎ {a})]

(∀M)P (M)

where the predicate P (M) stands M ∈ W (M(T ), <1). Therefore, we have to
prove:

1. ∅ ∈ W (M(T ), <1) (which is true by its definition).
2. (∀a)[(∀M)[M ∈ W (M(T ), <1) → M ⊎ {a} ∈ W (M(T ), <1)]].

This result is proved by well–founded induction on < (since it is well–
founded), with the following predicate P (a):

(∀M)[M ∈ W (M(T ), <1) → M ⊎ {a} ∈ W (M(T ), <1)]

With this, the proof is reduced to prove (∀b)[b < a → P (b)] → P (a),
or equivalently, to prove that if

(∀b)[b < a → (∀M)[M ∈ W (M(T ), <1) → M ⊎ {b} ∈ W (M(T ), <1)]]

then
(∀M)[M ∈ W (M(T ), <1) → M ⊎ {a} ∈ W (M(T ), <1)] (3)

We finish proving (3) using the weak induction axiom for the well–founded
part W (M(T ), <1), with the predicate Q(M):

M ⊎ {a} ∈ W (M(T ), <1) ∨ M /∈ W (M(T ), <1).

Finally, as consequence of lemma 2, we obtain the main theorem of this
section:

Theorem 3 (Dershowitz and Manna). Let < be a transitive and well–founded

relation on T . Then the relation <mult is a well–founded relation on M(T ).

less_mult_is_wf: THEOREM

transitive?[T](<) IMPLIES well_founded?[finite_bag[T]](less_mult)

A sketch of the proof is as follows. First, we show that the relation <mult

is contained in <+

1 . Furthermore, if the relation < is transitive, then <mult is
transitive and contains the relation <+

1 . Then, if < is transitive, <+

1 =<mult and
therefore, by lemma 2 and theorem 2, we conclude that <mult is well–founded.

3 Tableau Reasoning for ALC –satisfiability

In this section, we first describe the basic components of ALC logic, and we show
below how we have formalized in PVS the tableau based algorithm for this logic,
as well as the proofs of its termination, soundness and completeness.

3.1 Syntax and semantics of the ALC logic

We start presenting a brief introduction to the ALC –logic, its syntax and its
semantics, along with the corresponding description of its specification in PVS.

Let NC be a set of concept names and NR be a set of role names. The
set of ALC –concepts is built inductively from these names as described by the
following grammar, where A ∈ NC and R ∈ NR

C ::= A | ¬C | C1 ⊓ C2 | C1 ⊔ C2 | ∀R.C | ∃R.C

The set of ALC –concepts can be represented in PVS as a recursive datatype,
using the mechanism for defining abstract datatypes [20], and specifying the
constructors, the accessors and the recognizers. When a datatype is typechecked
in PVS, a new theory is created providing axioms and inductions principles for
this datatype. In particular, this theory contains the relation subterm (specifying

the notion of subconcept) and the well–founded relation <<, that is useful to make
recursive definitions on concepts.

To introduce the assertional knowledge, let NI be a set of individual names.
Given individual names x, y ∈ NI, a concept C and a role name R, the expressions
x: C and (x, y): R are called assertional axioms. An ABox A is a finite set of
assertional axioms. We specify in PVS the assertional axioms by a datatype and
the ABox by a type.

assertional_ax: DATATYPE

BEGIN

instanceof(left:NI, right:alc_concept) : instanceof?

related(left:NI, role:NR, right:NI) : related?

END assertional_ax

ABox: TYPE = finite_set[assertional_ax]

The semantics of description logics is defined in terms of interpretations. An
ALC –interpretation I is a pair I = (∆I , ·I), where ∆I is a non–empty set called
the domain, and ·I is an interpretation function that maps every concept name
A to a subset AI of ∆I , every role name R to a binary relation RI over ∆I and
every individual x to an element of ∆I . We represent in PVS an interpretation
I as a structure that contains the domain of I and the functions that define the
interpretation of concept names, role names, and the individuals.

interpretation: NONEMPTY_TYPE =

[# int_domain: (nonempty?[U]),

int_names_concept: [NC -> (powerset(int_domain))],

int_names_roles: [NR -> PRED[[(int_domain),(int_domain)]]],

int_names_ind: [NI -> (int_domain)] #]

It should be noted that in this specification we have used a universal type U

to represent the elements of the domain. Also, we have taken advantage of the
ability of PVS to deal with dependent types.

The interpretation function is extended to non-atomic concepts as follows

(¬D)I = ∆I \ DI

(C1 ⊓ C2)
I = CI

1 ∩ CI
2

(C1 ⊔ C2)
I = CI

1 ∪ CI
2

(∀R.D)I = {a ∈ ∆I : (∀b ∈ ∆I)[(a, b) ∈ RI → b ∈ DI ]}
(∃R.D)I = {a ∈ ∆I : (∃b ∈ ∆I)[(a, b) ∈ RI ∧ b ∈ DI ]}

We have specified this notion in PVS in a natural way, by recursion on C, using
the well–founded relation <<.

The interpretation I is a model of a concept C if CI 6= ∅. Thus, a concept
C is called satisfiable if it has a model

is_model_concept(I,C): bool = nonempty?(int_concept(C,I))

concept_satisfiable?(C): bool = EXISTS I: is_model_concept(I,C)

The interpretation I satisfies the assertional axiom x : C if xI ∈ CI and
satisfies (x, y): R if (xI , yI) ∈ RI . It satisfies the ABox A if it satisfies every
axiom in A. In that case, A is called satisfiable and I is called a model of A.

We have made the PVS formalization of the above definitions using the PVS
set theory and its capability of managing the existential and universal quantifiers.

The previous definitions naturally pose some standard inference problems
for DLs systems, such as concept and ABox satisfiability. It can be proved, and
we have done it in PVS, that concept satisfiability can be reduced to ABox
satisfiability (i.e., C is satisfiable iff for all x ∈ NI, {x:C} is satisfiable).

3.2 Deciding concept satisfiability for ALC

A tableau algorithm for ALC tries to prove the satisfiability of a concept C by
attempting to explicitly construct a model of C. This is done considering an
individual name x0 and manipulating the initial ABox {x0:C}, applying a set of
completion rules. In this process, we consider concepts in negation normal form
(NNF), a form in which negations appear only in front of concept names. This
does not mean a restriction since it is easy to specify a PVS function such that,
for each ALC –concept computes another equivalent 2 in NNF form.

An ABox A contains a clash if, for some individual name x ∈ NI and concept
name A ∈ NC, {x:A, x:¬A} ⊆ A. Otherwise, A is called clash–free. To test the
satisfiability of an ALC –concept C in NNF, the ALC –algorithm works starting
from the initial ABox {x0:C} and iteratively applying the following completion

rules

→⊓: if x:C ⊓ D ∈ A and {x:C, x:D} 6⊆ A
then A →⊓ A ∪ {x:C, x:D}

→⊔: if x:C ⊔ D ∈ A and {x:C, x:D} ∩ A = ∅
then A →⊔ A ∪ {x:E} for some E ∈ {C, D}

→∃: if x:∃R.D ∈ A and there is no y with {(x, y):R, y:D} ⊆ A
then A →∃ A ∪ {(x, y):R, y:D} for a fresh individual y

→∀: if x:∀R.D ∈ A and there is a y with (x, y):R ∈ A and y:D 6∈ A
then A →∀ A ∪ {y:D}

It stops when a clash has been generated or when no rule is applicable. In
the last case, the ABox is complete and a model can be derived from it. The
algorithm answers “C is satisfiable” if a complete and clash–free ABox has been
generated.

These completion rules can be seen as a production system. Nevertheless, we
have not formalized them in a functional way, but following a more declarative
style, defining the completion rules in PVS as binary relations between ABoxes.
For example, A1 →⊓ A2 if there exists an assertional axiom x:C ⊓D in A1 such
that {x:C, x:D} 6⊆ A1 and A2 = A1 ∪ {x:C, x:D}

and_step(AB1, AB2): bool =

EXISTS Aa: member(Aa,AB1) AND

instanceof?(Aa) AND

alc_and?(right(Aa)) AND

LET x = left(Aa), C = conc1(right(Aa)), D = conc2(right(Aa))

2 Two concepts are equivalent if they have the same models.

IN (NOT member(instanceof(x, C), AB1) OR

NOT member(instanceof(x, D), AB1)) AND

AB2 = add(instanceof(x, C), add(instanceof(x, D), AB1))

Once the rules have been specified in this way, we define the successor relation
on the ABoxes type: A1 → A2 if A1 does not contain a clash and A2 is obtained
from A1 by the application of a completion rule

successor(AB2,AB1): bool =

(NOT contains_clash(AB1)) AND

(and_step(AB1,AB2) OR or_step_1(AB1,AB2) OR or_step_2(AB1,AB2) OR

some_step(AB1,AB2) OR all_step(AB1,AB2))

It should be noted that we have specified the non-deterministic rule →⊔ by two
binary relations (or_step_1 and or_step_2), one for each component.

Take into account that the completion process can be seen as a closure pro-
cess, we say the ABox A2 is an expansion of the ABox A1 if A1

∗
→ A2, where

∗
→ is the reflexive and transitive closure of →

is_expansion(AB1)(AB2): bool = rtr_cl(successor)(AB2, AB1)

To illustrate the completion process, the following example shows the appli-
cation of some completion rules to an initial ABox {x0:C}

Example 1. Let C be the concept ∀R.D ⊓ (∃R.(D ⊔ E) ⊓ ∃R.(D ⊔ F )). Then,

A0 := {x0:∀R.D ⊓ (∃R.(D ⊔ E) ⊓ ∃R.(D ⊔ F ))
∗
→ A1 := A0 ∪ {x0:∀R.D, x0:∃R.(D ⊔ E), x0:∃R.(D ⊔ F )}
→ A2 := A1 ∪ {(x0, x1):R, x1:D ⊔ E}
→ A3 := A2 ∪ {x1:D}

Once defined the expansion relation, we use it to specify the notion of con-
sistency: An ABox A is consistent if it has a complete and clash–free expansion.
Similarly, a concept C is consistent if the initial ABox {x0:C} is consistent

is_consistent_abox(AB): bool =

EXISTS AB1: is_expansion(AB)(AB1) AND complete_clash_free(AB1)

is_consistent_concept(C): bool =

is_consistent_abox(singleton(instanceof(x_0,C)))

where complete_clash_free(A) holds if the ABox A is both complete and
clash–free.

This definition is the PVS specification of a generic ALC –algorithm for de-
ciding satisfiability of ALC –concepts. It should be pointed out the two kinds of
non–determinism in it: the way in which the rule →⊔ is applied (“don’t know”
non-determinism); and the choice of which rule to apply in each step and to
which axiom (“don’t care” non-determinism). To prove that the algorithm is
correct, we have to establish its termination, soundness and completeness. The
following subsections describe the corresponding PVS proofs.

3.3 Soundness

The ALC –algorithm is sound, that is:

alc_soundness: THEOREM

is_consistent_concept(C) IMPLIES concept_satisfiable?(C)

The PVS proof is based on the following steps:

1. If A is a complete and clash–free expansion of an initial ABox A0, then
A is satisfiable. We have proved this by specifying in PVS the canonical
interpretation IA associated with A, and proving that IA is a model of A.

2. If A1 → A2 and A2 is satisfiable, then A1 is satisfiable too.
3. Finally, we have proved the soundness theorem, using the induction scheme

suggested by the above definition of the expansion relation (which in turn is
based on the inductive definition of closure).

3.4 Termination

In order to verify the termination of the ALC –algorithm, it suffices to prove that
the successor relation, defined on the set E(C) of the expansions of the initial
ABox {x0:C}, is well founded

well_founded_successor: THEOREM

well_founded?[expansion_abox_concept(C)](successor)

where by the type expansion_abox_concept(C), we specify the set E(C)

expansion_abox_concept(C:(is_nnf?)): TYPE =

(is_expansion(singleton(instanceof(x_0, C))))

The proof of the well–foundedness of the successor relation is based on the
embedding lemma (Lemma 1). So, it suffices to show the existence of a type T ,
a well–founded relation < on T , and a function

µC : E(C) → T (1)

such that
(∀A1,A2)[A1 → A2 ⇒ µC(A2) < µC(A1)] (2)

Those functions with these properties are called measure functions.
The formalization of this proof in PVS has been carried out in two phases.

In the first one, we assume the existence of a measure function, including T and
< in the parameters, and (1) and (2) in the body of the PVS theory,

[..., T: TYPE+, <: (well_founded?[T])]: THEORY

measure_concept(C): [expansion_abox_concept(C) -> T]

measure_concept_decrease_successor: AXIOM

FORALL (AB1,AB2: expansion_abox_concept(C)):

successor(AB2,AB1)

IMPLIES measure_concept(C)(AB2) < measure_concept(C)(AB1)

and we prove that successor is well–founded on E(C).
In the second one, we prove the existence of measure functions. For the sake

of completeness, we outline Nutt’s definition of a measure function. In [18], W.
Nutt constructs a measure function taking T as the type of the finite multisets
of pairs of natural numbers M(N × N), and the well–founded order < as the
extension to M(N × N) of the lexicographic order on N × N, that we denote by
<mult. In order to formalize in PVS that construction, it is necessary carry out
two tasks:

1. To prove that the extension to M(N×N) of the lexicographic order on N×N

is a well–founded ordering. For this, it suffices to instantiate, in theorem 3,
T by N × N and < by the lexicographic ordering on N × N.

2. To define a function µC mapping each expansion A ∈ E(C) to a multiset of
pairs, such that µC(A2) <mult µC(A1) if A1 → A2. For the sake of clarity,
we devote section 4 to explain the details of its formalization.

3.5 Completeness

The last verification task is the proof of completeness of ALC –algorithm, that
is:

alc_completeness: THEOREM

concept_satisfiable?(C) IMPLIES is_consistent_concept(C)

The PVS proof is achieved by means three subtasks in turn:

1. If A is a satisfiable ABox, then A is clash–free.
2. If A1 is a satisfiable and not complete ABox, then there exists a satisfiable

ABox A2, which is successor of A1.
3. If A ∈ E(C) is satisfiable, then there exists a complete and clash–free expan-

sion of A in E(C) itself.

In order to carry out the second task, we have proved that for every satisfiable
not complete ABox A, for every rule r and for every axiom of A to which r
is applied, there exists an axiom such that by adding it we obtain a satisfiable
ABox. On the other hand, the last one is the key lemma for the main proof. We
have proved it by well–founded induction on the successor relation.

4 Measure on ALC–expansion of C

In this section we show a measure function on E(C) verifying the monotonicity
condition of section 3.4. The idea for defining the measure function µC is to map
each expansion A ∈ E(C) to a multiset of pairs, in such way that those pairs
represent all possibles rules that can be applied to A.

The first step is to define the notions of level and colevel. For this, we have
used the library of graphs of PVS [7]. We define the associated graph to an ABox
A, G(A), as the graph whose vertices are the individuals that occur in A, and
whose edges are the subsets {x, y}, such that (x, y):R ∈ A for some role R.

graph_assoc_abox(AB: ABox): graph[NI] =

(# vert:= occur_ni(AB), edges:= dbl_assoc_abox(AB) #)

From this definition, we prove that if A ∈ E(C), then G(A) is a tree with root
x0. This fact allows us to define the level of x in A as the length of the path
from x0 to x (minus 1), and the colevel of x in A, |x|A, as the difference between
the size of the concept C (denoted by |C|) and the level of x in A.

level(AB)(x:(occur_ni(AB))): nat = l(path_from_root(AB)(x)) - 1

colevel(AB)(x:(occur_ni(AB))): nat = size(C) - level(AB)(x)

Also, we prove that the colevel of an individual in A remains invariant under
the completion rules and that if y is a successor of x in A (i.e., (x, y):R ∈ A),
then |y|A = |x|A − 1. Both properties are essential to prove the monotonicity of
µC .

successor_preserve_colevel: LEMMA

occur_ni(AB1)(y) AND successor(AB2,AB1)

IMPLIES colevel(AB2)(y) = colevel(AB1)(y)

colevel_successor_related: LEMMA

successor_related(AB)(y,x) IMPLIES colevel(AB)(y) = colevel(AB)(x) - 1

As we have already said, the elements of the multiset associated to an ex-
pansion A should represent all possibles applicable rules to A. In some cases,
the applicability of a rule is completely determined by an instance axiom of A,
but that is not the case for the →∀ rule. Thus, in order to capture the notion of
applicability of a rule, we introduce the type activation (activ), whose elements
are structures consisting of an instance axiom and an individual, that made it
applicable. Then, we specify when an activation is applicable in A and we define
the agenda of A, agenda(A), as the set of applicable activations in A.

activ: TYPE = [# ax: (instanceof?), witness: NI #]

applicable_activ(Ac,AB): bool =

LET Aa = ax(Ac),y = witness(Ac),x = left(Aa),D = right(Aa) IN

member(Aa,AB) AND

CASES D OF

alc_a(A) : false,

alc_not(D1) : false,

alc_and(C1,C2) : x = y AND (NOT member(instanceof(x,C1),AB) OR

NOT member(instanceof(x,C2),AB)),

alc_or(C1,C2) : x = y AND NOT member(instanceof(x,C1),AB) AND

NOT member(instanceof(x,C2),AB),

alc_all(R,D1) : x /= y AND member(related(x,R,y),AB) AND

NOT member(instanceof(y,D1),AB),

alc_some(R,D1) : x = y AND NOT (EXISTS y: member(related(x,R,y),AB) AND

member(instanceof(y,D1),AB))

ENDCASES

agenda(AB): finite_set[activ] = {Ac | applicable_activ(Ac, AB)}

To specify the function µC we found the following difficulty: we can not define
a multiset in PVS in a declarative way, as with sets. Thus, the measure of the
expansion A, µC(A), is constructed by recursion in the agenda of A, adding the
pair (|y|A, |D|) for each applicable activation [x:D, y].

bag_assoc_activ(Ac, AB): finite_bag[[nat, nat]] =

IF NOT applicable_activ(Ac, AB) THEN emptybag

ELSE LET Aa = ax(Ac), y = witness(Ac), D = right(Aa) IN

singleton_bag((colevel(AB)(y), size(D)))

ENDIF

expansion_measure_aux(AB,(AB1: finite_set[activ])):

RECURSIVE finite_bag[[nat, nat]] =

IF empty?(AB1) THEN emptybag

ELSE plus(bag_assoc_activ(choose(AB1), AB),

expansion_measure_aux(AB, rest(AB1)))

ENDIF

MEASURE card(AB1)

expansion_measure(AB): finite_bag[[nat, nat]] =

expansion_measure_aux(AB, agenda(AB))

We illustrate the evolution of measures through the effect of completion rules
to example 1.

Example 2. The agendas and measures of the ABoxes of Example 1 are:

agenda measure

A0 {(x0:∀R.D ⊓ (∃R.(D ⊔ E) ⊓ ∃R.(D ⊔ F )), x0)} {̇(15, 15)}̇

A1 {(x0:∃R.(D ⊔ E), x0), (x0:∃R.(D ⊔ F ), x0)} {̇(15, 5), (15, 5)}̇

A2 {(x0:∃R.(D ⊔ F ), x0), (x0:∀R.D, x1), (x1:D ⊔ E, x1)} {̇(14, 3), (14, 3), (15, 5)}̇

A3 {(x0:∃R.(D ⊔ F ), x0)} {̇(15, 5)}̇

Finally, we prove the theorem that assures the monotony of µC .

Theorem 4. Let A1,A2 ∈ E(C). If A1 → A2 then µC(A2) <mult µC(A1) .

expansion_measure_decrease_successor: THEOREM

successor(AB2, AB1)

IMPLIES less_mult(expansion_measure(AB2), expansion_measure(AB1))

The formalization of the proof of this theorem in PVS has turned out to be
more difficult than the hand proof presented in [18]. Firstly, we can observe (in
Example 2) that if A1,A2 ∈ E(C) are such that A1 → A2, then there exists
an activation Ac1 = [x : D, y] ∈ agenda(A1), that matches with the applied
rule. In addition, Ac1 6∈ agenda(A2) and, for each activation Ac2 introduced
in the agenda(A2) as result of the rule application, its associated pair is smaller
(lexicographically) than (|y|A, |D|). Indeed, one of the following cases may occur:

1. Ac2 = [x:E, z], being z a successor of y in A2. Then, |z|A2
< |y|A2

= |y|A1
.

So, (|z|A2
, |E|) < (|y|A1

, |D|).

2. Ac2 = [x:E, y], being E a subconcept of D. In this case, |y|A2
= |y|A1

and
|E| < |D|. So, (|y|A2

, |E|) < (|y|A1
, |D|).

Secondly, we should note that the application of a rule can disable some acti-
vations of the agenda and, so, it can eliminate its associated pairs of the multiset.
Thus, we define in PVS the multiset K1 = µaux(A1, agenda(A1) \ agenda(A2)),
whose elements are the pairs associated to disabled activations. Also, the multi-
set K2 = µaux(A2, agenda(A2) \ agenda(A1)) that contains the pairs associated
to new enabled activations. Finally, M0 = µaux(A1, agenda(A1) ∩ agenda(A2))
is the multiset whose elements are the pairs associated to the activations that
remains enabled after the application of the rule. Regarding these multisets,
we prove the following properties: (1) K1 6= ∅, (2) µC(A2) = M0 ⊎ K2, (3)
µC(A1) = M0 ⊎ K1 y (4) (∀a ∈ K2)(∃b ∈ K1)[a < b]. Thus, we conclude that
µC(A2) <mult µC(A1).

Once the measure function has been constructed, the parameters T and <,
and the signature measure-concept(C) of subsection 3.4 are interpreted by the
appropriated mechanism of PVS.

5 Conclusions and future work

We have presented a formalization of the ALC logic in PVS, and a formaliza-
tion of a tableau–based algorithm for checking satisfiability of ALC –concepts,
proving its soundness, completeness and termination. Both for the termination
proof and the completeness proof we have used the well–foundedness of the
multiset relation induced by a well–founded relation, a property that we have
also formally proved. This last result have been proved in a general setting. So,
the parameters of the PVS theory can be instantiated in order to establish the
well–foundedness of concrete relations.

It should be pointed out that the choice of PVS as our verification system has
turned out to be beneficial for our formalization, since PVS provide definition of
abstract datatypes, inductive sets and dependent types, as well as parameterized
theories of sets, multisets and graphs.

To the best of our knowledge, it is the first work on formalizing DL reasoning,
although related works (see, for example [14, 11, 21]) have been done for other
logics. And, also, the first work in PVS about the well–foundedness of multiset
relations. Others formal proofs of the well–foundedness of the multiset ordering
have been carried out, in a similar way, in Coq [8] and in Isabelle. In [23], an
ACL2 formalization of the same result is also presented.

Finally, we point out some possible lines for future work. We plan to continue
the work following two research lines. First, we will apply type and operator re-
finement techniques presented in [3], in order to construct ALC –reasoners as
PVSio specifications, executable and formally verified, refining the specification
of the ALC –algorithm that we have presented here. Due to “don’t care” non-
determinism of the algorithm specified in PVS, its correctness will be translated
to specific implementations, whose efficiency will depend of the applied opti-
mization strategy.

Also, we are interested in extending the formalization of the ALC logic to
other descriptive logics, incrementally approaching us to the description logic
SHOIN , which is the description logic corresponding to OWL–DL [12], by
adding new constructors. The modular characteristic of the extensions of our
formalization may provide the automatic synthesis of ad–hoc reasoning systems
for specifics ontologies [2].

